Законы сохранения для КдФ

К лекции 2 (2024)

1 Определение D_x, D_x^{-1} и D_t

Это можно реализовать по разному. Мы поступим просто – адаптируем встроенные функции D, Derivative и Integrate, применив функции-обёртки.

Пусть u[n] обозначает $\partial_x^n(u(x))$. Производная по x и интеграл:

In[1]:=
```
n2x = u[n_] :→ Derivative[n][u][x];
x2n = {u[x] → u[0], Derivative[n_][u][x] :→ u[n]};

dx[expr_] := D[expr /. n2x, x] /. x2n
dx[expr_, k_] := D[expr /. n2x, {x, k}] /. x2n

int[expr_] := Block[{i = Integrate[expr /. n2x, x]},
  If[FreeQ[i, Integrate], i /. x2n, i]]
```

Тест. Если выражение не есть полная производная, интеграл от него не вычисляется.

In[6]:=
```
dx[u[1]^3 u[2] + u[4]]
dx[u[1]^3 u[2] + u[4], 2]
int[%]
int[u[1]^2]
```

Out[6]= $3\,u[1]^2\,u[2]^2 + u[1]^3\,u[3] + u[5]$

Out[7]= $6\,u[1]^2\,u[2]\times u[3] + u[2]\left(6\,u[1]\,u[2]^2 + 3\,u[1]^2\,u[3]\right) + u[1]^3\,u[4] + u[6]$

Out[8]= $3\,u[1]^2\,u[2]^2 + u[1]^3\,u[3] + u[5]$

Out[9]= $\int u'[x]^2\,dx$

Производная по t в силу уравнения $u_t = f$:

In[10]:=
```
n2t = u[n_] :→ Derivative[n, 0][u][x, t];
t2n = {u[x, t] → u[0], Derivative[n_, 0][u][x, t] :→ u[n]};

dt[expr_, f_] := D[expr /. n2t, t] /. u^(n_,1)[x, t] :→ D[f /. n2t, {x, n}] /. t2n
```

In[13]:=
```
dt[u[1] × u[2], u[3] + u[2]^4]
```

Out[13]=

$u[2]\left(4\,u[2]^3\,u[3] + u[4]\right) + u[1]\left(12\,u[2]^2\,u[3]^2 + 4\,u[2]^3\,u[4] + u[5]\right)$

2 Первые четыре закона сохранения

Определяем правую часть уравнения

In[14]:=
```
ut = u[3] + 6 u[0] × u[1]
```
Out[14]=
```
6 u[0] × u[1] + u[3]
```

Проверяем формулы из лекции

In[15]:=
```
Expand[dt[u[0], ut]]
int[%]
```
Out[15]=
```
6 u[0] × u[1] + u[3]
```
Out[16]=
```
3 u[0]^2 + u[2]
```

In[17]:=
```
Expand[dt[u[0]^2, ut]]
int[%]
```
Out[17]=
```
12 u[0]^2 u[1] + 2 u[0] × u[3]
```
Out[18]=
```
4 u[0]^3 - u[1]^2 + 2 u[0] × u[2]
```

In[19]:=
```
Expand[dt[u[1]^2 - 2 u[0]^3, ut]]
int[%]
```
Out[19]=
```
-36 u[0]^3 u[1] + 12 u[1]^3 + 12 u[0] × u[1] × u[2] - 6 u[0]^2 u[3] + 2 u[1] × u[4]
```
Out[20]=
```
-9 u[0]^4 + 12 u[0] u[1]^2 - 6 u[0]^2 u[2] - u[2]^2 + 2 u[1] × u[3]
```

In[21]:=
```
Expand[dt[u[2]^2 - 10 u[0] u[1]^2 + 5 u[0]^4, ut]]
int[%]
```
Out[21]=
```
120 u[0]^4 u[1] - 180 u[0] u[1]^3 - 120 u[0]^2 u[1] × u[2] + 36 u[1] u[2]^2 +
 20 u[0]^3 u[3] - 10 u[1]^2 u[3] + 12 u[0] × u[2] × u[3] - 20 u[0] × u[1] × u[4] + 2 u[2] × u[5]
```
Out[22]=
```
24 u[0]^5 - 90 u[0]^2 u[1]^2 + 10 (2 u[0]^3 + u[1]^2) u[2] +
 16 u[0] u[2]^2 - 20 u[0] × u[1] × u[3] - u[3]^2 + 2 u[2] × u[4]
```

3 Метод неопределенных коэффициентов

Генерация однородных многочленов

vars[f, u] список переменных u_n в выражении

mon[d, r, k] список мономов степени d от u_k, u_{k+1}, ... с общей суммой индексов равной r

hom[m, n, M] список всех мономов заданного веса M относительно веса $u_j \sim m + j\, n$

rhom[m, n, M] то же самое, но без мономов, линейных по старшей переменной (следовательно, все мономы в этом списке неэквивалентны по модулю Im D_x)

In[23]:=
```
vars[f_, u__] := Union[Flatten[Cases[f, Blank[#], {0, Infinity}] & /@ {u}]]

mon[d_, r_, k_] := If[d == 1, If[r ≥ k, {u[r]}, {}],
  Flatten[Table[u[s] × mon[d - 1, r - s, s], {s, k, r}]]]

hom[m_, n_, M_] := Flatten[Table[
   If[IntegerQ[(M - m d) / n], mon[d, (M - m d) / n, 0], {}],
   {d, 1, Floor[M / m]}]]

rhom[m_, n_, M_] := Select[hom[m, n, M], Exponent[#, Last[vars[#, u]]] > 1 || # == u[0] &]
```

In[27]:=
```
Table[rhom[2, 1, k], {k, 2, 13}] // TableForm
```

Out[27]//TableForm=

$u[0]$						
$u[0]^2$						
$u[1]^2$	$u[0]^3$					
$u[2]^2$	$u[0]\, u[1]^2$	$u[0]^4$				
$u[1]^3$						
$u[3]^2$	$u[0]\, u[2]^2$	$u[0]^2 u[1]^2$	$u[0]^5$			
$u[1]\, u[2]^2$	$u[0]\, u[1]^3$					
$u[4]^2$	$u[0]\, u[3]^2$	$u[2]^3$	$u[0]^2 u[2]^2$	$u[1]^4$	$u[0]^3 u[1]^2$	$u[0]^6$
$u[1]\, u[3]^2$	$u[0] \times u[1]\, u[2]^2$	$u[0]^2 u[1]^3$				

Вычисление плотности веса 8

Составим плотность и поток:

In[28]:=
```
ut = u[3] + 6 u[0] × u[1];

rhom[2, 1, 8]
r = {a, b, c}.%

hom[2, 1, 10]
s = Table[k[i], {i, 1, Length[%]}].%
```

Out[29]=

$\{u[2]^2, u[0]\, u[1]^2, u[0]^4\}$

Out[30]=

$c\, u[0]^4 + b\, u[0]\, u[1]^2 + a\, u[2]^2$

Out[31]=

$\{u[8], u[0] \times u[6], u[1] \times u[5], u[2] \times u[4], u[3]^2, u[0]^2 u[4],$
$u[0] \times u[1] \times u[3], u[0]\, u[2]^2, u[1]^2 u[2], u[0]^3 u[2], u[0]^2 u[1]^2, u[0]^5\}$

Out[32]=

$k[12]\, u[0]^5 + k[11]\, u[0]^2 u[1]^2 + k[10]\, u[0]^3 u[2] + k[9]\, u[1]^2 u[2] +$
$k[8] \times u[0]\, u[2]^2 + k[7] \times u[0] \times u[1] \times u[3] + k[5]\, u[3]^2 + k[6]\, u[0]^2 u[4] +$
$k[4] \times u[2] \times u[4] + k[3] \times u[1] \times u[5] + k[2] \times u[0] \times u[6] + k[1] \times u[8]$

Составим уравнение $r_t = s_x$, соберем коэффициенты, решим систему

In[33]:=
```
eq = dt[r, ut] - dx[s]

vars[eq, u]

Union[Flatten[CoefficientList[eq, %]]]

sol = Solve[% == 0]
```

Out[33]=

$$-5 k[12] u[0]^4 u[1] - 2 k[11] \times u[0] u[1]^3 - 3 k[10] u[0]^2 u[1] \times u[2] - 2 k[11] u[0]^2 u[1] \times u[2] - k[8] \times u[1] u[2]^2 - 2 k[9] \times u[1] u[2]^2 - k[10] u[0]^3 u[3] - k[7] u[1]^2 u[3] - k[9] u[1]^2 u[3] - k[7] \times u[0] \times u[2] \times u[3] - 2 k[8] \times u[0] \times u[2] \times u[3] + 4 c u[0]^3 (6 u[0] \times u[1] + u[3]) + b u[1]^2 (6 u[0] \times u[1] + u[3]) - 2 k[6] \times u[0] \times u[1] \times u[4] - k[7] \times u[0] \times u[1] \times u[4] - k[4] \times u[3] \times u[4] - 2 k[5] \times u[3] \times u[4] + 2 b u[0] \times u[1] \left(6 u[1]^2 + 6 u[0] \times u[2] + u[4]\right) - k[6] u[0]^2 u[5] - k[3] \times u[2] \times u[5] - k[4] \times u[2] \times u[5] + 2 a u[2] (18 u[1] \times u[2] + 6 u[0] \times u[3] + u[5]) - k[2] \times u[1] \times u[6] - k[3] \times u[1] \times u[6] - k[2] \times u[0] \times u[7] - k[1] \times u[9]$$

Out[34]=

{u[0], u[1], u[2], u[3], u[4], u[5], u[6], u[7], u[9]}

Out[35]=

$$\{0, -k[1], -k[2], -k[2] - k[3], 2 a - k[3] - k[4], -k[4] - 2 k[5], -k[6], 2 b - 2 k[6] - k[7], 12 a - k[7] - 2 k[8], 36 a - k[8] - 2 k[9], b - k[7] - k[9], 4 c - k[10], 18 b - 2 k[11], 12 b - 3 k[10] - 2 k[11], 24 c - 5 k[12]\}$$

Out[36]=

$$\left\{\left\{a \to \frac{c}{5}, b \to -2 c, k[1] \to 0, k[2] \to 0, k[3] \to 0, k[4] \to \frac{2 c}{5}, k[5] \to -\frac{c}{5}, k[6] \to 0, k[7] \to -4 c, k[8] \to \frac{16 c}{5}, k[9] \to 2 c, k[10] \to 4 c, k[11] \to -18 c, k[12] \to \frac{24 c}{5}\right\}\right\}$$

Проверка

In[37]:=
```
r8 = r /. sol[[1]] /. c → 5
s10 = s /. sol[[1]] /. c → 5

Expand[dt[r8, ut] - dx[s10]]
```

Out[37]=

$$5 u[0]^4 - 10 u[0] u[1]^2 + u[2]^2$$

Out[38]=

$$24 u[0]^5 - 90 u[0]^2 u[1]^2 + 20 u[0]^3 u[2] + 10 u[1]^2 u[2] + 16 u[0] u[2]^2 - 20 u[0] \times u[1] \times u[3] - u[3]^2 + 2 u[2] \times u[4]$$

Out[39]=

0

Вычисление в процедуре

Сделаем то же самое в виде процедуры для любого веса. Для однозначности, фиксируем коэффициент при квадрате старшей производной.

In[40]:=
```
ut = u[3] + 6 u[0] × u[1];

consl[m_] := Module[{r, s, a, b, eq},
  r = rhom[2, 1, m];
  r = Table[a[i], {i, 1, Length[r]}].r /. a[1] → 1;
  s = hom[2, 1, m + 2];
  s = Table[b[i], {i, 1, Length[s]}].s;
  eq = dt[r, ut] - dx[s];
  eq = Union[Flatten[CoefficientList[eq, vars[eq, u]]]];
  eq = Solve[eq == 0];
  If[Length[eq] == 0, {{0, 0}}, {r, s} /. eq] /. u[n_] :> u_n
 ]
```

In[42]:=
```
consl[2]
consl[3]
consl[4]
consl[5]
consl[6]
consl[7]
consl[8]
consl[9]
consl[10]
consl[11]
consl[12]
```

Out[42]=

$\{\{u_0, 3 u_0^2 + u_2\}\}$

Out[43]=

$\{\{0, 0\}\}$

Out[44]=

$\{\{u_0^2, 4 u_0^3 - u_1^2 + 2 u_0 u_2\}\}$

Out[45]=

$\{\{0, 0\}\}$

Out[46]=

$\{\{-2 u_0^3 + u_1^2, -9 u_0^4 + 12 u_0 u_1^2 - 6 u_0^2 u_2 - u_2^2 + 2 u_1 u_3\}\}$

Out[47]=

$\{\{0, 0\}\}$

Out[48]=

$\{\{5 u_0^4 - 10 u_0 u_1^2 + u_2^2, 24 u_0^5 - 90 u_0^2 u_1^2 + 20 u_0^3 u_2 + 10 u_1^2 u_2 + 16 u_0 u_2^2 - 20 u_0 u_1 u_3 - u_3^2 + 2 u_2 u_4\}\}$

Out[49]=

$\{\{0, 0\}\}$

Out[50]=

$\{\{-14 u_0^5 + 70 u_0^2 u_1^2 - 14 u_0 u_2^2 + u_3^2, -70 u_0^6 + 560 u_0^3 u_1^2 + 35 u_1^4 - 70 u_0^4 u_2 - 140 u_0 u_1^2 u_2 - 154 u_0^2 u_2^2 - 2 u_2^3 + 140 u_0^2 u_1 u_3 + 28 u_1 u_2 u_3 + 20 u_0 u_3^2 - 28 u_0 u_2 u_4 - u_4^2 + 2 u_3 u_5\}\}$

Out[51]=

$\{\{0, 0\}\}$

Out[52]=

$\{\{42 u_0^6 - 420 u_0^3 u_1^2 - 35 u_1^4 + 126 u_0^2 u_2^2 + 20 u_2^3 - 18 u_0 u_3^2 + u_4^2, 216 u_0^7 - 3150 u_0^4 u_1^2 - 840 u_0 u_1^4 + 252 u_0^5 u_2 + 1260 u_0^2 u_1^2 u_2 + 1176 u_0^3 u_2^2 + 462 u_1^2 u_2^2 + 156 u_0 u_2^3 - 840 u_0^3 u_1 u_3 - 140 u_1^3 u_3 - 504 u_0 u_1 u_2 u_3 - 234 u_0^2 u_3^2 - 18 u_2 u_3^2 + 252 u_0^2 u_2 u_4 + 60 u_2^2 u_4 + 36 u_1 u_3 u_4 + 24 u_0 u_4^2 - 36 u_0 u_3 u_5 - u_5^2 + 2 u_4 u_6\}\}$

4 Преобразование Миуры

Пусть v удовлетворяет уравнению мКдФ$^-$ $v_t = v_{xxx} - 6\left(v^2 + \lambda\right) v_x$, тогда переменная $u = v_x + v^2 + \lambda$ удовлетворяет уравнению КдФ $u_t = u_{xxx} - 6\, u\, u_x$.

Чтобы проверить это, *временно поменяем ролями u* и v (так как правила дифференцирования у нас заданы для u):

In[53]:=
```
ut = u[3] - 6 (u[0]^2 + λ) u[1];
v = u[1] + u[0]^2 + λ;
-dt[v, ut] + dx[v, 3] - 6 v dx[v]
Expand[%]
```

Out[55]=
```
12 u[0] u[1]^2 + 6 (λ + u[0]^2) u[2] + 6 u[1] × u[2] -
 6 (λ + u[0]^2 + u[1]) (2 u[0] × u[1] + u[2]) + 2 u[0] × u[3] - 2 u[0] (-6 (λ + u[0]^2) u[1] + u[3])
```

Out[56]=
```
0
```

5 Обращение преобразования Миуры

Положим $4\,\lambda = -z^2$:

$$v^2 + v_x = z^2/2 + u.$$

Подставим сюда разложение

$$v = -z/2 + V_0 + V_1/z + V_2/z^2 + \ldots$$

(индекс у V_k это просто номер, не производная), это даст

$$z^2/4 - z\,V_0 + \left(-V_1 + V_0^2\right) + (-V_2 + 2\,V_0\,V_1)/z$$
$$+ \left(-V_3 + 2\,V_0\,V_2 + V_1^2\right)/z^2 + \ \ldots \ + V_{0,x} + V_{1,x}/z + V_{2,x}/z^2 + \ldots \quad = \ z^2/4 + u.$$

Отсюда следует, что $V_0 = 0$, $V_1 = -u$, а на остальные коэффициенты получаем рекуррентные соотношения

$$V_{n+1} = V_1\,V_{n-1} + \ldots + V_{n-1}\,V_1 + V_{n,x} = 0, \quad n = 1, 2, 3, \ \ldots .$$

Вычислим несколько первых коэффициентов (скажем, 13, хотя машина может гораздо больше).

In[57]:=
```
M = 13;
V[1] = -u[0];
Do[V[n + 1] = Factor[(dx[V[n]] + Sum[V[s] × V[n - s], {s, 1, n - 1}])], {n, 1, M - 1}];
```

In[60]:=
```
Do[Print[V_n, "= ", V[n] /. u[k_] :→ u_k], {n, 1, M}]
```

$V_1 = -u_0$

$V_2 = -u_1$

$V_3 = u_0^2 - u_2$

$V_4 = 4 u_0 u_1 - u_3$

$V_5 = -2 u_0^3 + 5 u_1^2 + 6 u_0 u_2 - u_4$

$V_6 = -16 u_0^2 u_1 + 18 u_1 u_2 + 8 u_0 u_3 - u_5$

$V_7 = 5 u_0^4 - 50 u_0 u_1^2 - 30 u_0^2 u_2 + 19 u_2^2 + 28 u_1 u_3 + 10 u_0 u_4 - u_6$

$V_8 = 64 u_0^3 u_1 - 60 u_1^3 - 216 u_0 u_1 u_2 - 48 u_0^2 u_3 + 68 u_2 u_3 + 40 u_1 u_4 + 12 u_0 u_5 - u_7$

$V_9 = -14 u_0^5 + 350 u_0^2 u_1^2 + 140 u_0^3 u_2 - 442 u_1^2 u_2 - 266 u_0 u_2^2 - 392 u_0 u_1 u_3 + 69 u_3^2 - 70 u_0^2 u_4 + 110 u_2 u_4 + 54 u_1 u_5 + 14 u_0 u_6 - u_8$

$V_{10} = -256 u_0^4 u_1 + 960 u_0 u_1^3 + 1728 u_0^2 u_1 u_2 - 1224 u_1 u_2^2 + 256 u_0^3 u_3 - 900 u_1^2 u_3 - 1088 u_0 u_2 u_3 - 640 u_0 u_1 u_4 + 250 u_3 u_4 - 96 u_0^2 u_5 + 166 u_2 u_5 + 70 u_1 u_6 + 16 u_0 u_7 - u_9$

$V_{11} = 42 u_0^6 - 2100 u_0^3 u_1^2 + 1105 u_1^4 - 630 u_0^4 u_2 + 7956 u_0 u_1^2 u_2 + 2394 u_0^2 u_2^2 - 1262 u_2^3 + 3528 u_0^2 u_1 u_3 - 5564 u_1 u_2 u_3 - 1242 u_0 u_3^2 + 420 u_0^3 u_4 - 1630 u_1^2 u_4 - 1980 u_0 u_2 u_4 + 251 u_4^2 - 972 u_0 u_1 u_5 + 418 u_3 u_5 - 126 u_0^2 u_6 + 238 u_2 u_6 + 88 u_1 u_7 + 18 u_0 u_8 - u_{10}$

$V_{12} = 1024 u_0^5 u_1 - 9600 u_0^2 u_1^3 - 11\,520 u_0^3 u_1 u_2 + 13\,560 u_1^3 u_2 + 24\,480 u_0 u_1 u_2^2 - 1280 u_0^4 u_3 + 18\,000 u_0 u_1^2 u_3 + 10\,880 u_0^2 u_2 u_3 - 9524 u_2^2 u_3 - 7000 u_1 u_3^2 + 6400 u_0^2 u_1 u_4 - 11\,140 u_1 u_2 u_4 - 5000 u_0 u_3 u_4 + 640 u_0^3 u_5 - 2720 u_1^2 u_5 - 3320 u_0 u_2 u_5 + 922 u_4 u_5 - 1400 u_0 u_1 u_6 + 658 u_3 u_6 - 160 u_0^2 u_7 + 328 u_2 u_7 + 108 u_1 u_8 + 20 u_0 u_9 - u_{11}$

$V_{13} = -132 u_0^7 + 11\,550 u_0^4 u_1^2 - 24\,310 u_0 u_1^4 + 2772 u_0^5 u_2 - 87\,516 u_0^2 u_1^2 u_2 - 17\,556 u_0^3 u_2^2 + 69\,006 u_1^2 u_2^2 + 27\,764 u_0 u_2^3 - 25\,872 u_0^3 u_1 u_3 + 33\,760 u_1^3 u_3 + 122\,408 u_0 u_1 u_2 u_3 + 13\,662 u_0^2 u_3^2 - 26\,322 u_2 u_3^2 - 2310 u_0^4 u_4 + 35\,860 u_0 u_1^2 u_4 + 21\,780 u_0^2 u_2 u_4 - 20\,922 u_2^2 u_4 - 30\,776 u_1 u_3 u_4 - 5522 u_0 u_4^2 + 10\,692 u_0^2 u_1 u_5 - 20\,376 u_1 u_2 u_5 - 9196 u_0 u_3 u_5 + 923 u_5^2 + 924 u_0^3 u_6 - 4270 u_1^2 u_6 - 5236 u_0 u_2 u_6 + 1582 u_4 u_6 - 1936 u_0 u_1 u_7 + 988 u_3 u_7 - 198 u_0^2 u_8 + 438 u_2 u_8 + 130 u_1 u_9 + 22 u_0 u_{10} - u_{12}$

Можно проверить, что построенный отрезок ряда действительно удовлетворяет уравнению (5), с точностью до членов z^n с $n \geq M$:

In[61]:=
```
v = -z / 2 + Sum[V[n] / z^n, {n, 1, M}];
Series[Expand[v^2 + dx[v]], {z, Infinity, M - 1}]
```

Out[62]=

$$\frac{z^2}{4} + u[0] + O\left[\frac{1}{z}\right]^{13}$$

Многочлены V_n являются плотностями з.с. для КдФ, то есть, существуют функции σ_n от u_k такие, что $D_t(V_n) = D_x(\sigma_n)$.

Короткий список плотностей:

In[63]:=
```
ut = u[3] - 6 u[0] × u[1];
rhos = Table[V[n], {n, 1, 10}]
```

Out[64]=
```
{-u[0], -u[1], u[0]^2 - u[2], 4 u[0] × u[1] - u[3],
 -2 u[0]^3 + 5 u[1]^2 + 6 u[0] × u[2] - u[4], -16 u[0]^2 u[1] + 18 u[1] × u[2] + 8 u[0] × u[3] - u[5],
 5 u[0]^4 - 50 u[0] u[1]^2 - 30 u[0]^2 u[2] + 19 u[2]^2 + 28 u[1] × u[3] + 10 u[0] × u[4] - u[6],
 64 u[0]^3 u[1] - 60 u[1]^3 - 216 u[0] × u[1] × u[2] -
  48 u[0]^2 u[3] + 68 u[2] × u[3] + 40 u[1] × u[4] + 12 u[0] × u[5] - u[7],
 -14 u[0]^5 + 350 u[0]^2 u[1]^2 + 140 u[0]^3 u[2] - 442 u[1]^2 u[2] - 266 u[0] u[2]^2 - 392 u[0] × u[1] × u[3] +
  69 u[3]^2 - 70 u[0]^2 u[4] + 110 u[2] × u[4] + 54 u[1] × u[5] + 14 u[0] × u[6] - u[8],
 -256 u[0]^4 u[1] + 960 u[0] u[1]^3 + 1728 u[0]^2 u[1] × u[2] - 1224 u[1] u[2]^2 +
  256 u[0]^3 u[3] - 900 u[1]^2 u[3] - 1088 u[0] × u[2] × u[3] - 640 u[0] × u[1] × u[4] +
  250 u[3] × u[4] - 96 u[0]^2 u[5] + 166 u[2] × u[5] + 70 u[1] × u[6] + 16 u[0] × u[7] - u[9]}
```

Дифференцируем его по t и находим σ интегрированием:

In[65]:=
```
Expand[dt[rhos, ut]];
sigmas = Expand[int[%]]
```

Out[66]=
```
{3 u[0]^2 - u[2], 6 u[0] × u[1] - u[3], -4 u[0]^3 + 5 u[1]^2 + 8 u[0] × u[2] - u[4],
 -24 u[0]^2 u[1] + 18 u[1] × u[2] + 10 u[0] × u[3] - u[5],
 9 u[0]^4 - 60 u[0] u[1]^2 - 42 u[0]^2 u[2] + 19 u[2]^2 + 28 u[1] × u[3] + 12 u[0] × u[4] - u[6],
 96 u[0]^3 u[1] - 60 u[1]^3 - 252 u[0] × u[1] × u[2] -
  64 u[0]^2 u[3] + 68 u[2] × u[3] + 40 u[1] × u[4] + 14 u[0] × u[5] - u[7],
 -24 u[0]^5 + 450 u[0]^2 u[1]^2 + 200 u[0]^3 u[2] - 442 u[1]^2 u[2] - 304 u[0] u[2]^2 - 448 u[0] × u[1] × u[3] +
  69 u[3]^2 - 90 u[0]^2 u[4] + 110 u[2] × u[4] + 54 u[1] × u[5] + 16 u[0] × u[6] - u[8],
 -384 u[0]^4 u[1] + 1080 u[0] u[1]^3 + 2160 u[0]^2 u[1] × u[2] - 1224 u[1] u[2]^2 +
  352 u[0]^3 u[3] - 900 u[1]^2 u[3] - 1224 u[0] × u[2] × u[3] - 720 u[0] × u[1] × u[4] +
  250 u[3] × u[4] - 120 u[0]^2 u[5] + 166 u[2] × u[5] + 70 u[1] × u[6] + 18 u[0] × u[7] - u[9],
 70 u[0]^6 - 2800 u[0]^3 u[1]^2 + 1105 u[1]^4 - 910 u[0]^4 u[2] + 8840 u[0] u[1]^2 u[2] +
  2926 u[0]^2 u[2]^2 - 1262 u[2]^3 + 4312 u[0]^2 u[1] × u[3] - 5564 u[1] × u[2] × u[3] - 1380 u[0] u[3]^2 +
  560 u[0]^3 u[4] - 1630 u[1]^2 u[4] - 2200 u[0] × u[2] × u[4] + 251 u[4]^2 - 1080 u[0] × u[1] × u[5] +
  418 u[3] × u[5] - 154 u[0]^2 u[6] + 238 u[2] × u[6] + 88 u[1] × u[7] + 20 u[0] × u[8] - u[10],
 1536 u[0]^5 u[1] - 11520 u[0]^2 u[1]^3 - 14976 u[0]^3 u[1] × u[2] + 13560 u[1]^3 u[2] +
  26928 u[0] × u[1] u[2]^2 - 1792 u[0]^4 u[3] + 19800 u[0] u[1]^2 u[3] +
  13056 u[0]^2 u[2] × u[3] - 9524 u[2]^2 u[3] - 7000 u[1] u[3]^2 + 7680 u[0]^2 u[1] × u[4] -
  11140 u[1] × u[2] × u[4] - 5500 u[0] × u[3] × u[4] + 832 u[0]^3 u[5] - 2720 u[1]^2 u[5] -
  3652 u[0] × u[2] × u[5] + 922 u[4] × u[5] - 1540 u[0] × u[1] × u[6] + 658 u[3] × u[6] -
  192 u[0]^2 u[7] + 328 u[2] × u[7] + 108 u[1] × u[8] + 22 u[0] × u[9] - u[11]}
```

Проверка:

In[67]:=
```
Expand[dt[rhos, ut] - dx[sigmas]]
```

Out[67]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Если посмотреть внимательно на выписанную выше таблицу, можно заметить, что все V_n с четными номерами сами являются полными производными по x, ещё до дифференцирования по t:

In[68]:=

```
Table[V[n], {n, 2, M, 2}]

int[%]
```

Out[68]=

```
{-u[1], 4 u[0] × u[1] - u[3], -16 u[0]^2 u[1] + 18 u[1] × u[2] + 8 u[0] × u[3] - u[5],
 64 u[0]^3 u[1] - 60 u[1]^3 - 216 u[0] × u[1] × u[2] - 48 u[0]^2 u[3] + 68 u[2] × u[3] + 40 u[1] × u[4] +
  12 u[0] × u[5] - u[7], -256 u[0]^4 u[1] + 960 u[0] u[1]^3 + 1728 u[0]^2 u[1] × u[2] - 1224 u[1] u[2]^2 +
  256 u[0]^3 u[3] - 900 u[1]^2 u[3] - 1088 u[0] × u[2] × u[3] - 640 u[0] × u[1] × u[4] +
  250 u[3] × u[4] - 96 u[0]^2 u[5] + 166 u[2] × u[5] + 70 u[1] × u[6] + 16 u[0] × u[7] - u[9],
 1024 u[0]^5 u[1] - 9600 u[0]^2 u[1]^3 - 11520 u[0]^3 u[1] × u[2] + 13560 u[1]^3 u[2] +
  24480 u[0] × u[1] u[2]^2 - 1280 u[0]^4 u[3] + 18000 u[0] u[1]^2 u[3] +
  10880 u[0]^2 u[2] × u[3] - 9524 u[2]^2 u[3] - 7000 u[1] u[3]^2 + 6400 u[0]^2 u[1] × u[4] -
  11140 u[1] × u[2] × u[4] - 5000 u[0] × u[3] × u[4] + 640 u[0]^3 u[5] - 2720 u[1]^2 u[5] -
  3320 u[0] × u[2] × u[5] + 922 u[4] × u[5] - 1400 u[0] × u[1] × u[6] + 658 u[3] × u[6] -
  160 u[0]^2 u[7] + 328 u[2] × u[7] + 108 u[1] × u[8] + 20 u[0] × u[9] - u[11]}
```

Out[69]=

```
{-u[0], 2 u[0]^2 - u[2], -16/3 u[0]^3 + 5 u[1]^2 + 8 u[0] × u[2] - u[4],
 16 u[0]^4 - 60 u[0] u[1]^2 - 48 u[0]^2 u[2] + 20 u[2]^2 + 28 u[1] × u[3] + 12 u[0] × u[4] - u[6],
 -256/5 u[0]^5 + 480 u[0]^2 u[1]^2 + 4 (64 u[0]^3 - 113 u[1]^2) u[2] - 320 u[0] u[2]^2 -
  448 u[0] × u[1] × u[3] + 69 u[3]^2 - 16 (6 u[0]^2 - 7 u[2]) u[4] + 54 u[1] × u[5] + 16 u[0] × u[6] - u[8],
 512 u[0]^6/3 - 3200 u[0]^3 u[1]^2 + 1130 u[1]^4 - 80 u[0] (16 u[0]^3 - 113 u[1]^2) u[2] +
  3200 u[0]^2 u[2]^2 - 3904 u[2]^3/3 + 20 u[1] (224 u[0]^2 - 281 u[2]) u[3] - 1380 u[0] u[3]^2 +
  40 (16 u[0]^3 - 41 u[1]^2 - 56 u[0] × u[2]) u[4] + 252 u[4]^2 - 2 (540 u[0] × u[1] - 209 u[3]) u[5] -
  80 (2 u[0]^2 - 3 u[2]) u[6] + 88 u[1] × u[7] + 20 u[0] × u[8] - u[10]}
```

Такие плотности не засчитываются, они отвечают *тривиальным* законам сохранения. А с нечетными номерами все в порядке, это нетривиальные плотности.

In[70]:=
```
Table[V[n], {n, 1, M, 2}]
int[%]
```

Out[70]=
```
{-u[0], u[0]^2 - u[2], -2 u[0]^3 + 5 u[1]^2 + 6 u[0] × u[2] - u[4],
 5 u[0]^4 - 50 u[0] u[1]^2 - 30 u[0]^2 u[2] + 19 u[2]^2 + 28 u[1] × u[3] + 10 u[0] × u[4] - u[6],
 -14 u[0]^5 + 350 u[0]^2 u[1]^2 + 140 u[0]^3 u[2] - 442 u[1]^2 u[2] - 266 u[0] u[2]^2 - 392 u[0] × u[1] × u[3] +
  69 u[3]^2 - 70 u[0]^2 u[4] + 110 u[2] × u[4] + 54 u[1] × u[5] + 14 u[0] × u[6] - u[8],
 42 u[0]^6 - 2100 u[0]^3 u[1]^2 + 1105 u[1]^4 - 630 u[0]^4 u[2] + 7956 u[0] u[1]^2 u[2] +
  2394 u[0]^2 u[2]^2 - 1262 u[2]^3 + 3528 u[0]^2 u[1] × u[3] - 5564 u[1] × u[2] × u[3] - 1242 u[0] u[3]^2 +
  420 u[0]^3 u[4] - 1630 u[1]^2 u[4] - 1980 u[0] × u[2] × u[4] + 251 u[4]^2 - 972 u[0] × u[1] × u[5] +
  418 u[3] × u[5] - 126 u[0]^2 u[6] + 238 u[2] × u[6] + 88 u[1] × u[7] + 18 u[0] × u[8] - u[10],
 -132 u[0]^7 + 11 550 u[0]^4 u[1]^2 - 24 310 u[0] u[1]^4 + 2772 u[0]^5 u[2] - 87 516 u[0]^2 u[1]^2 u[2] -
  17 556 u[0]^3 u[2]^2 + 69 006 u[1]^2 u[2]^2 + 27 764 u[0] u[2]^3 - 25 872 u[0]^3 u[1] × u[3] +
  33 760 u[1]^3 u[3] + 122 408 u[0] × u[1] × u[2] × u[3] + 13 662 u[0]^2 u[3]^2 -
  26 322 u[2] u[3]^2 - 2310 u[0]^4 u[4] + 35 860 u[0] u[1]^2 u[4] + 21 780 u[0]^2 u[2] × u[4] -
  20 922 u[2]^2 u[4] - 30 776 u[1] × u[3] × u[4] - 5522 u[0] u[4]^2 + 10 692 u[0]^2 u[1] × u[5] -
  20 376 u[1] × u[2] × u[5] - 9196 u[0] × u[3] × u[5] + 923 u[5]^2 + 924 u[0]^3 u[6] -
  4270 u[1]^2 u[6] - 5236 u[0] × u[2] × u[6] + 1582 u[4] × u[6] - 1936 u[0] × u[1] × u[7] +
  988 u[3] × u[7] - 198 u[0]^2 u[8] + 438 u[2] × u[8] + 130 u[1] × u[9] + 22 u[0] × u[10] - u[12]}
```

Out[71]=

$$\Big\{-\int u[x]\,dx,\ \int\big(u[x]^2-u''[x]\big)\,dx,\ \int\big(-2u[x]^3+5u'[x]^2+6u[x]\,u''[x]-u^{(4)}[x]\big)\,dx,$$

$$\int\big(5u[x]^4-50u[x]\,u'[x]^2-30u[x]^2u''[x]+19u''[x]^2+28u'[x]\,u^{(3)}[x]+10u[x]\,u^{(4)}[x]-u^{(6)}[x]\big)\,dx,$$

$$\int\big(-14u[x]^5+350u[x]^2u'[x]^2+140u[x]^3u''[x]-442u'[x]^2u''[x]-266u[x]\,u''[x]^2-392u[x]\,u'[x]\,u^{(3)}[x]+69u^{(3)}[x]^2-70u[x]^2u^{(4)}[x]+110u''[x]\,u^{(4)}[x]+54u'[x]\,u^{(5)}[x]+14u[x]\,u^{(6)}[x]-u^{(8)}[x]\big)\,dx,$$

$$\int\big(42u[x]^6-2100u[x]^3u'[x]^2+1105u'[x]^4-630u[x]^4u''[x]+7956u[x]\,u'[x]^2u''[x]+2394u[x]^2u''[x]^2-1262u''[x]^3+3528u[x]^2u'[x]\,u^{(3)}[x]-5564u'[x]\,u''[x]\,u^{(3)}[x]-1242u[x]\,u^{(3)}[x]^2+420u[x]^3u^{(4)}[x]-1630u'[x]^2u^{(4)}[x]-1980u[x]\,u''[x]\,u^{(4)}[x]+251u^{(4)}[x]^2-972u[x]\,u'[x]\,u^{(5)}[x]+418u^{(3)}[x]\,u^{(5)}[x]-126u[x]^2u^{(6)}[x]+238u''[x]\,u^{(6)}[x]+88u'[x]\,u^{(7)}[x]+18u[x]\,u^{(8)}[x]-u^{(10)}[x]\big)\,dx,$$

$$\int\big(-132u[x]^7+11550u[x]^4u'[x]^2-24310u[x]\,u'[x]^4+2772u[x]^5u''[x]-87516u[x]^2u'[x]^2u''[x]-17556u[x]^3u''[x]^2+69006u'[x]^2u''[x]^2+27764u[x]\,u''[x]^3-25872u[x]^3u'[x]\,u^{(3)}[x]+33760u'[x]^3u^{(3)}[x]+122408u[x]\,u'[x]\,u''[x]\,u^{(3)}[x]+13662u[x]^2u^{(3)}[x]^2-26322u''[x]\,u^{(3)}[x]^2-2310u[x]^4u^{(4)}[x]+35860u[x]\,u'[x]^2u^{(4)}[x]+21780u[x]^2u''[x]\,u^{(4)}[x]-20922u''[x]^2u^{(4)}[x]-30776u'[x]\,u^{(3)}[x]\,u^{(4)}[x]-5522u[x]\,u^{(4)}[x]^2+10692u[x]^2u'[x]\,u^{(5)}[x]-20376u'[x]\,u''[x]\,u^{(5)}[x]-9196u[x]\,u^{(3)}[x]\,u^{(5)}[x]+923u^{(5)}[x]^2+924u[x]^3u^{(6)}[x]-4270u'[x]^2u^{(6)}[x]-5236u[x]\,u''[x]\,u^{(6)}[x]+1582u^{(4)}[x]\,u^{(6)}[x]-1936u[x]\,u'[x]\,u^{(7)}[x]+988u^{(3)}[x]\,u^{(7)}[x]-198u[x]^2u^{(8)}[x]+438u''[x]\,u^{(8)}[x]+130u'[x]\,u^{(9)}[x]+22u[x]\,u^{(10)}[x]-u^{(12)}[x]\big)\,dx\Big\}$$

Итак, наш эксперимент приводит к следующим предположениям:

1) Все коэффициенты в разложении $v(z)$ являются плотностями з.с. для КдФ;

2) плотности с чётными номерами тривиальны;

3) плотности с нечётными номерами нетривиальны.